

On souhaite éditer une facture à l'écran, avec calcul automatique du prix TTC pour chaque ligne de la facture (en fonction du prix unitaire HT, de la quantité et du taux de TVA) et du montant total de la facture (somme des prix TTC calculés pour chaque ligne). Le taux de TVA peut être normal (20,6 %) ou réduit (5,5 %).

- 1) Créer une application conforme au modèle ci-dessous. Indiquer votre nom et votre prénom directement sur la feuille. Enregistrer le projet sur votre disquette en utilisant votre nom de famille comme nom des fichiers. Ne pas oublier de donner des noms explicites aux composants visuels qui seront utilisés dans le code.
- 2) Programmer le bouton « Nouvelle facture » de manière à vider les listes d'édition qui composent la facture, à remettre le montant total à zéro et à mettre à jour l'étiquette associée. On pourra utiliser la méthode *Clear* des composants *ListBox* pour effacer simplement les listes.
- 3) Programmer le bouton « Ajouter à la facture », puis vérifier vos calculs avec le jeu de données ci-dessous\*. À chaque clic sur le bouton, les opérations suivantes devront être exécutées :
  - détermination du taux de TVA ;
  - calcul du prix TTC pour la ligne de facture en cours ;
  - cumul du montant total de la facture ;
  - affichage de la ligne de facture dans les listes d'édition + affichage du montant total dans l'étiquette appropriée ;
  - réinitialisation des zones de texte pour la saisie (vides) et du taux de TVA (normal par défaut).
- 4) Programmer le bouton « Quitter » pour sortir de l'application.
- 5) Créer, dans un module, une fonction permettant d'ajouter derrière les prix affichés dans la facture (uniquement) l'unité de mesure « F ».

Désignation	Prix unitaire	Quantité	Prix total
PC Pentium III	10000 F	2	24120 F
Ecran 17"	2500 F	2	6030 F
Imprimante laser	4500 F	1	5427 F
Scanner	990 F	1	1133,94 F
Manuel d'utilisation	500 F	1	527,5 F

Total facture  
37298,44 F

Quitter

\* Seul le manuel d'utilisation bénéficie d'un taux de TVA réduit.

## Code du module

```
Function Monétaire(x As Single) As String
    Monétaire = x & " F"
End Function
```

## Code de la feuille

```
Dim TOTAL As Single

Private Sub Ajouter_Click()
    ' Déclarations
    Dim pu As Single
    Dim q As Single
    Dim p As Single
    Dim tx As Single
    ' Initialisations
    pu = PrixU_saisie.Text
    q = Qté_saisie.Text
    ' Détermination taux de TVA
    If TVAnormale Then
        tx = 0.206
    Else
        tx = 0.055
    End If
    ' Calcul du prix
    p = q * pu * (1 + tx)
    ' Cumul du total
    TOTAL = TOTAL + p
    ' Affichage
    Dési_facture.AddItem (Dési_saisie.Text)
    PrixU_facture.AddItem (Monétaire(pu))
    Qté_facture.AddItem (q)
    Prix_tot.AddItem (Monétaire(p))
    Total_facture.Caption = Monétaire(TOTAL)
    ' Réinitialisation des champs de saisie
    Dési_saisie.Text = ""
    PrixU_saisie.Text = ""
    Qté_saisie.Text = ""
    TVAnormale.Value = True
End Sub

Private Sub Nouvelle_Click()
    ' Réinitialisation de la facture
    Dési_facture.Clear
    PrixU_facture.Clear
    Qté_facture.Clear
    Prix_tot.Clear
    TOTAL = 0
    Total_facture.Caption = Monétaire(TOTAL)
End Sub

Private Sub Quitter_Click()
    End
End Sub
```